

NAG Fortran Library Routine Document

F07AJF (SGETRI/DGETRI)

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

F07AJF (SGETRI/DGETRI) computes the inverse of a real matrix A , where A has been factorized by F07ADF (SGETRF/DGETRF).

2 Specification

```
SUBROUTINE F07AJF(N, A, LDA, IPIV, WORK, LWORK, INFO)
ENTRY      sgetri (N, A, LDA, IPIV, WORK, LWORK, INFO)
INTEGER    N, LDA, IPIV(*), LWORK, INFO
real      A(LDA,*), WORK(*)
```

The ENTRY statement enables the routine to be called by its LAPACK name.

3 Description

To compute the inverse of a real matrix A , the routine must be preceded by a call to F07ADF (SGETRF/DGETRF), which computes the LU factorization of A as $A = PLU$. The inverse of A is computed by forming U^{-1} and then solving the equation $XPL = U^{-1}$ for X .

4 References

Du Croz J J and Higham N J (1992) Stability of methods for matrix inversion *IMA J. Numer. Anal.* **12** 1–19

5 Parameters

- 1: N – INTEGER *Input*
On entry: n , the order of the matrix A .
Constraint: $N \geq 0$.
- 2: A(LDA,*) – ***real*** array *Input/Output*
Note: the second dimension of the array A must be at least $\max(1, N)$.
On entry: the LU factorization of A , as returned by F07ADF (SGETRF/DGETRF).
On exit: the factorization is overwritten by the n by n matrix A^{-1} .
- 3: LDA – INTEGER *Input*
On entry: the first dimension of the array A as declared in the (sub)program from which F07AJF (SGETRI/DGETRI) is called.
Constraint: $LDA \geq \max(1, N)$.
- 4: IPIV(*) – INTEGER array *Input*
Note: the dimension of the array IPIV must be at least $\max(1, N)$.
On entry: the pivot indices, as returned by F07ADF (SGETRF/DGETRF).

5: WORK(*) – *real* array Workspace

Note: the dimension of the array WORK must be at least $\max(1, \text{LWORK})$.

On exit: if INFO = 0, WORK(1) contains the minimum value of LWORK required for optimum performance.

6: LWORK – INTEGER Input

On entry: the dimension of the array WORK as declared in the (sub)program from which F07AJF (SGETRI/DGETRI) is called, unless LWORK = -1, in which case a workspace query is assumed and the routine only calculates the optimal dimension of WORK (using the formula given below).

Suggested value: for optimum performance LWORK should be at least $N \times nb$, where *nb* is the **blocksize**.

Constraint: $\text{LWORK} \geq \max(1, N)$ or $\text{LWORK} = -1$.

7: INFO – INTEGER Output

On exit: INFO = 0 unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

Errors or warnings detected by the routine:

INFO < 0

If $\text{INFO} = -i$, the *i*th parameter had an illegal value. An explanatory message is output, and execution of the program is terminated.

INFO > 0

If $\text{INFO} = i$, the *i*th diagonal element of the factor *U* is zero, *U* is singular, and the inverse of *A* cannot be computed.

7 Accuracy

The computed inverse *X* satisfies a bound of the form:

$$|XA - I| \leq c(n)\epsilon|X|P|L||U|,$$

where $c(n)$ is a modest linear function of *n*, and ϵ is the **machine precision**.

Note that a similar bound for $|AX - I|$ cannot be guaranteed, although it is almost always satisfied. See Du Croz and Higham (1992).

8 Further Comments

The total number of floating-point operations is approximately $\frac{4}{3}n^3$.

The complex analogue of this routine is F07AWF (CGETRI/ZGETRI).

9 Example

To compute the inverse of the matrix *A*, where

$$A = \begin{pmatrix} 1.80 & 2.88 & 2.05 & -0.89 \\ 5.25 & -2.95 & -0.95 & -3.80 \\ 1.58 & -2.69 & -2.90 & -1.04 \\ -1.11 & -0.66 & -0.59 & 0.80 \end{pmatrix}.$$

Here *A* is nonsymmetric and must first be factorized by F07ADF (SGETRF/DGETRF).

9.1 Program Text

Note: the listing of the example program presented below uses *bold italicised* terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```
*      F07AJF Example Program Text
*      Mark 15 Release. NAG Copyright 1991.
*      .. Parameters ..
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
      INTEGER          NMAX, LDA, LWORK
      PARAMETER        (NMAX=8,LDA=NMAX,LWORK=64*NMAX)
*      .. Local Scalars ..
      INTEGER          I, IFAIL, INFO, J, N
*      .. Local Arrays ..
      real             A(LDA,NMAX), WORK(LWORK)
      INTEGER          IPIV(NMAX)
*      .. External Subroutines ..
      EXTERNAL         sgetrf, sgetri, X04CAF
*      .. Executable Statements ..
      WRITE (NOUT,*) 'F07AJF Example Program Results'
*      Skip heading in data file
      READ (NIN,*)
      READ (NIN,*) N
      IF (N.LE.NMAX) THEN
*
*          Read A from data file
*
          READ (NIN,*) ((A(I,J),J=1,N),I=1,N)
*
*          Factorize A
*
          CALL sgetrf(N,N,A,LDA,IPIV,INFO)
*
          WRITE (NOUT,*)
          IF (INFO.EQ.0) THEN
*
*              Compute inverse of A
*
              CALL sgetri(N,A,LDA,IPIV,WORK,LWORK,INFO)
*
*              Print inverse
*
              IFAIL = 0
              CALL X04CAF('General',' ',N,N,A,LDA,'Inverse',IFAIL)
          ELSE
              WRITE (NOUT,*) 'The factor U is singular'
          END IF
      END IF
      STOP
*
      END
```

9.2 Program Data

```
F07AJF Example Program Data
4                               :Value of N
1.80   2.88   2.05  -0.89
5.25  -2.95  -0.95  -3.80
1.58  -2.69  -2.90  -1.04
-1.11  -0.66  -0.59   0.80   :End of matrix A
```

9.3 Program Results

F07AJF Example Program Results

Inverse	1	2	3	4
1	1.7720	0.5757	0.0843	4.8155
2	-0.1175	-0.4456	0.4114	-1.7126
3	0.1799	0.4527	-0.6676	1.4824
4	2.4944	0.7650	-0.0360	7.6119
